

3.8. Funktionen mit Alternativen

In vielen Fällen ist leider vorher – beim Aufbau oder der Nutzung der Kalkulation – noch nicht klar, welche Formel man verwenden muss. Die Auswahl aus mindestens zwei möglichen Wegen (Alternativen) muss auf Grund einer (oder mehrerer) Bedingung(en) erfolgen. Mit den normalen Funktionen lässt sich so etwas nicht (oder nur mit erheblichen Aufwand) realisieren.

Allen Bedingungen in EXCEL-Alternativen ist gemeinsam, dass sie so formuliert werden müssen, dass man sie eindeutig mit **Ja** oder **Nein** beantworten werden können.



Q: www.flickr.com (eti: right.left.right)

Exakterweise sollte man vielleicht sagen mit **Wahr** oder **Falsch**. Die Bedingung trifft also zu oder nicht.


Bei einfachen Funktionen (einfache Alternativen) gibt es nur den Wahr-Zweig – d.h. die Funktion wird nur ausgeführt, wenn die Bedingung wahr ist (z.B. ZÄHLENWENN, SUMMEWENN).

Die Hauptalternativ-Funktion WENN lässt beide Zweige zu. Die einzelnen Alternativ-Zweige können auch leer sein. Praktischerweise kann man auch in jedem der Zweige weitere Alternativen einbauen.

Liegen Daten für die Nutzung in Alternativen in anderen Tabellen(-bereichen) vor, dann helfen die VERWEIS-Funktionen bei der Erstellung recht unkomplizierter Mehrfachalternativen.

3.8.1. einfache Alternativen (ZÄHLENWENN, SUMMEWENN)

Bei einfachen Alternativen wird nur die Bedingung geprüft und die Funktion dann ausgeführt, wenn die Bedingung wahr ist. Einen alternativen Zweig für die Ablehnung der Bedingung existiert nicht.



Will man z.B. die Anzahl von Einträgen wissen, die einem bestimmten Kriterium entsprechen, dann können wir hierfür die Funktion ZÄHLENWENN nutzen. Sie zählt nur weiter, wenn die angegebene Bedingung wahr ist.

ZÄHLENWENN(Zellbereich ; Bedingung)

In der folgenden Kalkulation soll die Anzahl der Einkäufe des Kunden Meier ermittelt werden.

	A	B	C	D
1				
2		Kunde	Einkaufs-Betrag	
3		Meier	23,62 €	
4		Lehmann	104,00 €	
5		Meier	98,50 €	
6		Meier	2,98 €	
7		Schulz	478,32 €	
8	Einkäufe Meier			
9				

Der Zellbereich, in dem gesucht werden soll (Suchbereich), ist leicht bestimmt (B2:B7). Die Bedingung für das Zählen ist die Übereinstimmung mit dem Text „Meier“. Somit lautet die Formel:

```
=ZÄHLENWENN(B3:B7; "Meier")
```

Wie erwartet - liefert uns EXCEL als Ergebnis die 3.
Nach dem gleichen Prinzip lassen sich die Einkäufe über 20 Euro zählen.

```
=ZÄHLENWENN(C3:C7; ">20")
```

Neben dem >-Zeichen (größer als) können auch das <- und das =-Zeichen benutzt werden. Sinnvolle Kombinationen wie >= und <= sind ebenfalls zugelassen. Dabei ist es egal, ob nach größeren oder kleineren Zahlen usw. oder Texten gesucht werden soll. Einträge, wie ">Meier" oder "<=Schulz" sind also gültig. Um die Bedeutung ungleich (≠) zu realisieren, kann <> verwendet werden.

Einen Teil eines Textes oder Wortes sucht man mit Hilfe von Wildcards bzw. Jokerzeichen. Das einfachste ist der Stern "*". Er steht für beliebig viele Buchstaben. Mit:

```
=ZÄHLENWENN(C3:C7; "M*")
```

werden alle Namen im Zellbereich C3:C7 gezählt, die mit M beginnen und dann beliebige Buchstaben enthalten. Ersetzt man den Bedingungsteil durch "**M***", dann werden alle Texte gezählt die ein M enthalten. Ob davor oder dahinter oder an beiden Seiten weitere Zeichen folgen ist egal. Das zweite Joker-Zeichen ist das Fragezeichen "?". Es steht für exakt ein Zeichen. Ansonsten gilt die gleiche Verwendung wie beim Stern.

Als Bedingungen sind auch Zelladressen zulässig. Die Anführungszeichen entfallen dann ersatzlos. Leider lassen sich keine logischen Funktionen wie UND, ODER und NICHT in die Bedingung einbauen.

Die Anzahl der Einkäufe von Lehmann könnte also auch so formuliert werden:

```
=ZÄHLENWENN(B3:B7; B4)
```

Die Übersichtlichkeit und Pflfegbarkeit einer Kalkulation sollte man bei der Verwendung solcher Konstrukte immer im Auge behalten.

Für eine bedingte Summenbildung verwendet man SUMMEWENN. Der Syntax lehnt sich an der Funktion ZÄHLENWENN an:

SUMMEWENN(Zellbereich ; Bedingung)

Ein praktische Verwendungszweck ist z.B. die Aufsummierung aller Einzahlungen (positive Werte) bzw. aller Auszahlungen (negative Werte) auf einem Konto.

Aufgabe A:

- 1. Lassen Sie in der obigen Beispieltabelle zählen, wie oft Lehmann bzw. Schulz eingekauft haben!*
- 2. Zählen Sie alle Einkäufe, die nicht von Schulz getätigt wurden!*
- 3. Berechnen Sie die Summe aller Einkäufe über 100 Euro!*
- 4. Bauen Sie die Kalkulation so um, dass in der Zelle C8 die Einkäufe von der Person stehen, die jeweils in Zelle B8 eingegeben wird!*

Aufgabe B:

- 1. Realisieren Sie einen einfachen Konto-Auszug der TabKalk-Bank! Auf dem Formular sollen maximal 10 Vorgänge / Buchungen möglich sein. In der Kalkulation soll der alte Kontostand als Eingabe möglich sein. Berechnet werden sollen:*
 - a) der neue Kontostand*
 - b) die Anzahl und der Gesamtbetrag der Einzahlungen / Zugänge*
 - c) die Anzahl und der Gesamtbetrag der Auszahlungen / Abgänge!*

Nun werden Sie vielleicht denken „Schade, dass ich nicht so einfach alle Einkäufe von Meier summieren kann!“ Weit gefehlt – auch dies ist möglich. EXCEL bietet dazu eine Erweiterung der SUMMEWENN-Funktion. Optional können wir nämlich hinter der Bedingung noch den Summierungsbereich angeben:

SUMMEWENN(Zellbereich ; Bedingung ; Summierungsbereich)

Fehlt der Summierungsbereich, dann wird im Suchbereich (Zellbereich) summiert. Das genannte Beispiel ließe sich dann so formulieren:

```
=SUMMEWENN(B2:B7; "Meier"; C2:C7)
```

Der Syntax der SUMMEWENN-Funktion läßt sich so dann notieren:

SUMMEWENN(Zellbereich ; Bedingung [; Summierungsbereich])

Aufgabe (Fortsetzung zu Aufgabe A):

- 1. Summieren Sie die Einkäufe von allen anderen Kunden (also z.B. nicht die von Meier)!*

3.8.2. echte Alternativen (WENN)

Bei echten Alternativen sind beide Zweige (Bedingung ist **wahr** oder **falsch**) vorgesehen. Man muss sie nicht in jedem Fall nutzen – man hat aber die Möglichkeit, es zu tun.

Typische "kaufmännische" Anwendungen sind z.B.:

- **WENN** Bestellung < Lagerbestand **DANN** Bestellung realisieren (verkaufen) **SONST** Lagerbestand auffüllen (ev. Rest verkaufen)
- **WENN** Kontostand > 0 **DANN** Spartzins aufaddieren **SONST** Kreditzins abziehen
- **WENN** Umsatz > 100000 € **DANN** Rabatt = 5% **SONST** Rabatt = 3%
- **WENN** Alter < 18 **DANN** Beitrag = 10 € **SONST** Beitrag = 30 €



Q: www.flickr.com (Toni Mitjanit: Vies de tren)

Praktischer Weise heißt die Funktion WENN. Zu Anfang ist der Syntax dieser Funktion etwas gewöhnungsbedürftig.

WENN(Bedingung ; [Dann-Anweisung] [; [Sonst-Anweisung]])


Den Aufbau und die Schreibung von Bedingungen haben wir bei den einfachen Alternativen (→ [3.8.1. einfache Alternativen \(ZÄHLENWENN, SUMMEWENN\)](#)) schon aufgezeigt. Die Dann- bzw. Sonst-Anweisungen können sehr verschiedengestaltig sein. Mit Formeln (ohne beginnendes Gleichheitszeichen) lassen sich Berechnungen durchführen. Bei Angabe von Texten werden diese angezeigt. Und letztendlich sind auch wieder WENN-Funktionen zulässig.

In der Praxis muß man die unterschiedlichen zulässigen Notierungen der WENN-Funktion beachten. Dies bezieht sich besonders auf die optionale Sonst-Anweisung. Es ist sowohl zulässig das Semikolon nach der Dann-Anweisung wegzulassen bzw. mitzuschreiben. Dies hat aber praktische Auswirkungen! Wird das Semikolon mitgeschrieben, dann wird bei leeren Sonst-Anweisung im Eintrittsfall eine 0 zurückgeliefert.

Ohne Semikolon erhalten wir ein FALSCH.

Betrachten wir nun ein Rabatt-Beispiel:

Ein Kunde bekommt beim Kauf von 3 gleichen Artikeln einen als Rabatt. Zuerst einmal wollen wir uns nur anzeigen lassen, ob es für einen bestimmten Artikel den Rabatt gibt oder nicht.



	A	B	C	D	E	F
1	Rechnung					
2						
3	Artikel	Anzahl	Einzel-Verkaufspreis	Rabatt		
4	Milch 1,5%	5	0,80 €			
5	Fein-Brot	2	1,80 €			
6	Brötchen	10	0,20 €			
7	Apfel	3	0,30 €			
8						

Bringen wir die Bedingung und die "Ergebnisse" zuerst in einem Struktogramm unter:

Anzahl > 2 ?	
ja (wahr)	nein (falsch)
es gibt Rabatt	es gibt KEINEN Rabatt

Nun formulieren wir hieraus die Alternativaussage für die Spalte D (Rabatt) umgangssprachlich:

- **WENN** Anzahl > 2 **DANN** es gibt Rabatt (ja) **SONST** kein Rabatt (nein)

In die WENN-Funktion eingesetzt und in typischer Formelschreibweise lautet der Eintrag für die Zelle D4:

=WENN(B4>2; "ja"; "nein")

Wir vereinfachen nun den Rabatt auf 33,33% ab 3 Artikel. Der Rabatt soll – statt **ja/nein**-Aussage – nun als Geldbetrag angezeigt werden. Weiterhin soll noch der Gesamt-Verkaufspreis berechnet werden.

	A	B	C	D	E	F
1	Rechnung					
2						
3	Artikel	Anzahl	Einzel-Verkaufspreis	Rabatt	Gesamt-Verkaufspreis	
4	Milch 1,5%	5	0,80 €	ja		
5	Fein-Brot	2	1,80 €	nein		
6	Brötchen	10	0,20 €	ja		
7	Apfel	3	0,30 €	ja		
8						

Formulieren wir zuerst wieder die Alternativaussage für die Spalte D (Rabatt) umgangssprachlich:

- **WENN** Anzahl > 2 **DANN** Rabatt = Anzahl * Einzel-Verkaufspreis * 33,33%
SONST Rabatt = 0

Daraus lässt sich die folgende Formel für D4 ableiten:

```
=WENN(B4>2; B4*C4*33,33%; 0)
```

Für den Verkaufspreis gibt es verschiedene Lösungen – auch einige, die ohne WENN-Funktion auskommen. Wir wählen hier aber eine mit, da wir ja gerade diese kennen lernen wollen: (Dann könnten wir später auch auf die Rabatt-Spalte verzichten.)

- **WENN** Anzahl > 2 **DANN** Gesamt-Verkaufspreis = Anzahl * Einzel-Verkaufspreis * (1 - 33,33%) **SONST** Gesamt-Verkaufspreis = Anzahl * Einzel-Verkaufspreis

Die passende Formel für E4:

```
=WENN(B4>2; B4*C4*(1-33,33%); B4*C4)
```

Letztendlich brauchen wir nur noch die Summierung der rabattierten Verkaufspreise organisieren und fertig ist unsere Rechnung:

	A	B	C	D	E	F
1	Rechnung					
2						
3	Artikel	Anzahl	Einzel-Verkaufspreis	Rabatt	Gesamt-Verkaufspreis	
4	Milch 1,5%	5	0,80 €	1,32 €	2,68 €	
5	Fein-Brot	2	1,80 €	0 €	1,80 €	
6	Brötchen	10	0,20 €	0,67 €	1,33 €	
7	Apfel	3	0,30 €	0,30 €	0,60 €	
8			Summen:	2,29 €	6,41 €	
9						

Denkaufgabe:
Mit welcher Formel / Rechnung kann man den Gesamtverkaufspreis ohne eine WENN-Funktion aus den anderen Spalten berechnen?

Aufgaben:


1. Ändern Sie die Rabatt-Kalkulation so, dass es beim Kauf von 5 Stücken das 5. als Rabatt gibt! Berechnen Sie auch die Rechnungssumme!
2. Prüfen Sie, ob bei der WENN-Funktionen zur Rabatt-Berechnung der eigentlich ungenutzte SONST-Zweig weggelassen werden kann! Welche Wirkungen hat diese Änderung auf die Anzeige der Rabatte?
3. Eine Bank gibt bei Einlagen von über 5000 Euro einen Zins von 3% p.a. im Vergleich zu sonst 1,75%. Berechnen Sie den Zins und den neuen Kontostand für die Kunden A, B, C, ... bei folgenden Kontoständen (im gesamten letzten Jahr):
1200 €, 7435 €, 3000 €, 5000 €, 302 €, 0 €, 1000 €, 5001 €, 2998 €
4. Die genannten Kunden haben den gleichen Konto-Betrag auf einem anderen Konto als einjährigen Kredit mit einem Zinssatz p.a. von 8% ab 3000 Euro und sonst 9,2%. Berechnen Sie den Kreditzins und den aktuellen Kontostand!

für die gehobene Anspruchsebene:

5. Ändern Sie die Kalkulation von Aufgabe 1 so ab, das nur noch beim Kauf von jeweils 5 Stück der Rabatt gewährt wird! (z.B.: bei 13 Stück wird der Rabatt 2x gewährt, die restlichen 3 Stück werden ohne Rabatt verkauft!)

3.8.2.1. Schachtelung von Alternativen - Mehrfachauswahl

Der **Dann-** und / oder der **Sonst-Zweig** einer WENN-Funktion kann wiederum vollständige WENN-Funktionen enthalten. Man spricht von der Schachtelung der Funktion. Jede WENN-Funktion muss für sich gesehen geschlossen sein, d.h. sie darf nicht durch Teile einer anderen WENN-Funktion unterbrochen sein, die dann ev. später fortgesetzt wird.



Einige mögliche Verzweigungs-Konstrukte sind in der folgenden Tabelle aufgezeigt. Die einzelnen Entwicklungsreihen lassen sich beliebig weiter fortsetzen oder kombinieren.

Zur deutlichen Unterscheidung sind in den folgenden Beispielen die einzelnen Wenn-Funktionen jeweils in einer anderen Farbe angezeigt. Für **Bed** können immer jeweils passende Bedingungen eingesetzt werden. Die Platzhalter **Dann** und **Sonst** stehen für beliebige passende Formeln, Funktionen oder Ausgaben:

zulässige Beispiel-Konstrukte	
1	=WENN(Bed; Dann; Sonst)
2	=WENN(Bed; WENN(Bed; Dann; Sonst); Sonst)
3	=WENN(Bed; WENN(Bed; WENN(Bed; Dann; Sonst); Sonst); Sonst)
4	=WENN(Bed; WENN(Bed; WENN(Bed; WENN(Bed; Dann; Sonst); Sonst); Sonst); Sonst)
	...
5	=WENN(Bed; Dann; WENN(Bed; Dann; Sonst))
6	=WENN(Bed; Dann; WENN(Bed; Dann; WENN(Bed; Dann; Sonst)))
7	=WENN(Bed; Dann; WENN(Bed; Dann; WENN(Bed; Dann; WENN(Bed; Dann; Sonst))))
	...
8	=WENN(Bed; WENN(Bed; Dann; Sonst); WENN(Bed; Dann; Sonst))
9	=WENN(Bed; WENN(Bed; WENN(Bed; Dann; Sonst); Sonst); WENN(Bed; Dann; Sonst))
10	=WENN(Bed; WENN(Bed; Dann; Sonst); WENN(Bed; Dann; WENN(Bed; Dann; Sonst)))
11	=WENN(Bed; WENN(Bed; Dann; WENN(Bed; Dann; Sonst)); WENN(Bed; Dann; Sonst))

Besonders die Konstrukte 4 und 7 sind typische Mehrfachalternativen. Schritt für Schritt – bzw. WENN- für WENN-Funktion – werden bestimmte Bedingungen ausgeschlossen bis am Ende nur noch eine Sonst-Anweisung (für alle Rest-Fälle) übrig bleibt. Solche Konstrukte lassen sich schrittweise aufbauen und testen. In langen – in einem Arbeitsgang eingegebenen – Konstrukten ist es oft sehr schwer einen Fehler zu finden.

Aufgaben:

1. Erstellen Sie zu den Konstrukten 2, 6, 8 und 11 die Blockschemata!
2. Prüfen Sie, ob die nachfolgenden Konstrukte theoretisch zulässig wären! Begründen Sie Ihre Prüfergebnisse! Weisen Sie auf ev. Probleme der Konstrukte bei einer praktischen Verwendung hin!

(Bed .. gültige Bedingung; Dann, Sonst .. gültiger Ausdruck)

a	=WENN(Bed; Dann; WENN(Bed; Dann))
b	=WENN(Bed; WENN(Bed; WENN(Dann; Sonst); Sonst); Sonst)
c	=WENN(Bed, WENN(Bed, WENN(Bed, Dann, Sonst), Sonst), Sonst)
d	=WENN(Bed; WENN(Bed; Dann))
e	=WENN(Dann; WENN(Bed; Dann; Sonst); Sonst WENN(Bed; Dann; Sonst))
f	=WENN(Bed, WENN(Bed; ; Sonst); WENN(Bed; ; Sonst))
g	=WENN(Bed; WENN(Bed; WENN(Bed; Dann; Sonst; Sonst; Sonst)))
h	=WENN(WENN(Bed; Dann; Sonst); Dann; WENN(Bed; Dann; Sonst)))
i	=WENN(Bed; WENN(Dann; WENN(Bed; Dann; Sonst; Dann; Sonst)))
j	=WENN(WENN(Dann; Sonst); Dann; WENN(Bed; Dann; Sonst)))
k	=WENN(Bed;;)

Prinzipiell gibt keine beste Vorgehensweise für den Aufbau von Mehrfach- oder verzweigten Alternativen. Manchmal ist es besser, Sonderfälle gleich zu Anfang auszusortieren (z.B. die mögliche Division durch Null). In anderen Fällen ist es einfacher, mit möglichst groben Bedingungen gleich viele Fälle auf einem Mal zu bearbeiten bzw. die Fälle sinnvoll (sachlich) zu zerteilen (dies entspricht ja auch dem Prinzip der Alternativen). In jedem Fall ist es aber günstig, verschachtelte Alternativen schrittweise zu entwickeln und zu testen.

Als praktische Beispiel nehmen wir uns die Berechnung eines neuen Kontostandes vor. Die Beispielbank gibt ab 2000 Euro Einlage 2% Zinsen p.a.. Bei Saldo wird bis 1000 Euro ein Dispositionskredit von 4% p.a. eingeräumt. Der normale Kreditzinssatz ist 6% p.a. Wie muss nun die Berechnungsformel für den neuen Kontostand lauten, wenn wir von einem über ein Jahr eingefrorenen Kontostand ausgehen?

Beim Analysieren der Aufgabenstellung wird sofort klar, dass es zwei grundsätzlich zu unterscheidende Fälle gibt. Entweder das Konto ist im Plus (oder gleich Null) und es gibt eventuell Sparzinsen, oder das Konto ist im Minus und es muss mit Kreditzinsen belastet werden.

Kontostand >= 0 ?	
ja (wahr)	nein (falsch)
sparen (möglich)	kreditieren

Im Spar-Fall wird leider erst ab 2000 Euro ein Zins gewährt. Also muss auch hier eine Unterscheidung vorgenommen werden.

Kontostand >= 0 ?		
ja (wahr)		nein (falsch)
Kontostand >= 2000		kreditieren
ja (wahr)	nein (falsch)	
2 % zuaddieren	Kontostand übernehmen	

Als Letztes müssen wir noch die Kreditbedingungen umsetzen. Bis –1000 Euro haben wir einen günstigeren Dispositionskredit mit nur 4% Kreditzins. Ist der Kontostand kleiner, dann kassiert die Bank den Normalzins.

Kontostand ≥ 0 ?			
ja (wahr)		nein (falsch)	
Kontostand ≥ 2000		Kontostand < -1000	
ja (wahr)	nein (falsch)	ja (wahr)	nein (falsch)
2% zuaddieren	Kontostand übernehmen	6% abziehen	4% abziehen

Setzen wir die entwickelten Struktogramme nun genauso schrittweise in EXCEL-Formeln um. Für die Adressen sei die folgende Tabelle gegeben:

	A	B	C
1	Kontostandberechnung		
2			
3	Jahres-Kontostand	neuer Kontostand	
4	734,93 €		
5			

Die erste Alternative setzen wir probeweise so um, dass uns die WENN-Funktion einen Text ausgibt.

```
=WENN(A4>0; "sparen"; "kreditieren")
```

	A	B	C
1	Kontostandberechnung		
2			
3	Jahres-Kontostand	neuer Kontostand	
4	734,93 €	sparen	
5			

Nun ist schon die erste Gelegenheit unsere Alternative ausgiebig zu testen. Dazu sollte man sich schon vorher eine Testreihe von Kontoständen anlegen. Diese sollte mindestens die Grenzwerte und jeweils Werte über und unten den Grenzwerten enthalten. Gute Testreihen enthalten weiterhin zufällige Werte.

Als Nächstes verfeinern wir den Sparen-Zweig. Wir ersetzen also die Textausgabe "sparen" durch die Alternative: (neuer Konstrukt zur besseren Unterscheidung farbig)

WENN(A4 \geq 2000; A4*(1+2%); A4)

Das ergibt zusammen:

```
=WENN(A4>0; WENN(A4 $\geq$ 2000; A4*(1+2%); A4); "kreditieren")
```

Beim Testen erhalten wir nun im Spar-Fall schon immer einen Euro-Betrag angezeigt. Der Kredit-Fall bleibt noch als Text.

	A	B	C
1	Kontostandberechnung		
2			
3	Jahres-Kontostand	neuer Kontostand	
4	734,93 €	734,93 €	
5			

Es folgt die Alternative für den Kredit-Fall:

WENN(A4<-1000; A4*(1+4%); A4*(1+6%))

Nach dem Einsetzen erhalten wir die endgültige Formel:

=WENN(A4>0; WENN(A4>=2000; A4*(1+2%); A4); **WENN(A4<-1000; A4*(1+4%); A4*(1+6%))**)

Zum Testen wurde auch noch der ehemalige Kontostand negiert, damit wir ein neues Ergebnis erhalten:

	A	B	C
1	Kontostandberechnung		
2			
3	Jahres-Kontostand	neuer Kontostand	
4	-734,93 €	-779,03 €	
5			

Aufgaben A:

1. Eine Bank gibt bei Einlagen von über 5000 Euro einen Zins von 3% p.a. im Vergleich zu sonst 1,75%. Berechnen Sie den Zins und den neuen Kontostand für die Kunden A, B, C, ... bei folgenden Kontoständen (im gesamten letzten Jahr):
1200 €, 7435 €, 3000 €, 5000 €, 302 €, 0 €, 1000 €, 5001 €, 2998 €, 3001 €, 4999 €, 698 €, 10000 € ; 100 €
2. Ändern Sie jetzt die Kontostände der Kunden B, C, E, I, J und M auf den negativen Wert! Prüfen Sie Ihre Kalkulation auf finanztechnisch richtiges Rechnen aus der Sicht der Bank! Passen Sie die Kalkulation gegebenenfalls an das beobachtete Problem an!
3. Arbeiten Sie nun das Kreditwesen ein! Normalerweise gibt es einen jährlichen Kreditzins von 8 %. Als Dispo-Kredit wird bei weniger als 3000 Euro (Schuld) ein Zinssatz von 6,2 % angesetzt.

Aufgaben B:

1. Die Bank ändert ihre Konditionen: „Sparen über 4000 Euro mit 2,5% p.a., sonst 1,5%. Dispokredit unter 1500 Euro zu 5% p.a. und ansonsten 9,5%“. Passen Sie die Kalkulation so an, dass die neuen Bedingungen umgesetzt werden!

für die gehobene Anspruchsebene:

2. Bauen Sie die Kalkulation so um, dass ein Konstanten-Bereich mit den Konditionen der Bank verfügbar ist!

für FREAKS:

3. Vereinfachen Sie die Formeln so, dass man mit 3 Vergleichen und mit 2 Rechenoperationen (+, -, *, /) auskommt (statt bisher 8)!

Aufgaben C:

In einer Lagerkartei soll der Lagerstatus für jeden einzelnen Artikel angezeigt werden! Als Meldebestand wurde der doppelte Mindestbestand festgelegt. Bestellzeiten brauchen nicht beachtet werden. Der Lagerleiter wünscht folgende Texte:

"voll"	solange keine Bestelleinheit mehr dazu passen würde	"nachbestellen"	wenn der Meldebestand erreicht wurde
"i.O."	zwar nachbestellt werden könnte, aber noch nicht muß	"kritisch"	wenn der Bestand unter dem Minimum ist

1. Berechnen Sie alle notwendigen Werte (Lagerkennzahlen)!

2. Lassen Sie sich zu jedem Artikel den Lagerstatus anzeigen!

für die gehobene Anspruchsebene:

3. Realisieren einen farbigen Hintergrund des Lagerstatus entsprechend der Vorgabe!

Aufgaben D:

Aus Effektivitätsgründen sollen ab sofort nur noch Monats-Kataloge an Kunden geschickt werden, die in den letzten 12 Monaten etwas bestellt haben. Zur "Erinnerung" wird im letzten Monat vor der Streichung als Katalogkunde noch ein Sonderangebotskatalog verschickt.

1. Erstellen Sie eine Kalkulation, die für jeden Kunden berechnet, wie viele Monate und wie viele Tage (insgesamt) seit der letzten Bestellung vergangen sind!

für xy bitte das Vorjahr nehmen

Kunde	letzte Bestellung	Kunde	letzte Bestellung	Kunde	letzte Bestellung
Meier	14.08.20xy	Volk	05.06.20xy	Arendt	25.02.20xy
Düse	12.01.20xy	Kern	03.09.20xy	Zander	04.07.20xy
Klein	23.08.20xy	Müller	30.03.20xy	Dorn	31.10.20xy
Kieser	01.05.20xy	Meyer	04.07.20xy	Hauser	18.04.20xy
Holt	04.04.20xy	Rossow	20.12.20xy	Schmidt	18.11.20xy

2. Lassen Sie sich in einer weiteren Spalte anzeigen, welchen Katalog der Kunde bei der nächsten Lieferung geschickt bekommen muß

Aufgaben E:

Der Vertriebsleiter möchte für die 5 Vertreter Ihrer Firma das Provisionssystem umstellen. Derzeit muß die Firma wirtschaftlich etwas kürzer treten. Ihre Firma zahlt normalerweise 4 % Provision und bei Umsätzen über 20000 Euro 5 % Provision (auf den gesamten Umsatz eines Vertreters).

1. Berechnen Sie in einer Kalkulation die Provisionen für die Vertreter entsprechend ihren Umsätzen!

Vertreter	Vertriebsgebiet	Umsatz
Herr Schulz	Nord	35000 €
Frau Meier	Mitte	9000 €
Frau von Schütt	West	10000 €
Herr Niedecker	Süd	20000 €
Frau Penz	Ost	37500 €

2. Kalkulieren Sie nun zusätzlich das neue Provisionssystem mit den gleichen Umsatzzahlen! Der Vertriebsleiter sieht für jeden Vertreter eine umsatzunabhängige Grundprovision von 500 Euro vor. Bei Umsätzen ab 10000 Euro erhält der Vertreter zusätzlich 1 % Provision, ab 20000 Euro 2 % und über 35000 Euro 4 % Provision (immer auf den gesamten Umsatz des Vertreters bezogen).
3. Formulieren und begründen Sie in einem Textfeld Ihren Standpunkt zum neuen Provisionssystem!
4. Gestalten Sie alles so, dass der Ausdruck nur aus einer A4-Seite besteht! Drucken Sie die Seite einmal aus!
5. Warten Sie vor einem weiteren Ausdruck auf neue Umsatzzahlen! (Die Aussagen im Textfeld können für diesen Ausdruck gelöscht werden bzw. werden (vom Kursleiter) nicht beachtet.)

für die gehobene Anspruchsebene:

6. Arbeiten Sie mit einer Definitionsregion / einem Konstantenbereich und lassen Sie EXCEL automatisch anzeigen, ob das neue System billiger oder teurer für die Firma ist!

für FREAKS:

7. Erstellen Sie neben den anderen Kalkulationen eine weitere, in der die Provision jedes Vertreters abgestuft berechnet wird! (Also (leicht abgewandelt) z.B. für die ursprüngliche Variante: bis 10000 Euro gibt es 4 %, für jeden Euro über 10000 Euro 5 % Provision usw.)
8. Wie ist diese Version aus der Sicht der Firma einzuschätzen? Nutzen Sie dazu ein weiteres Textfeld!

3.8.2.2. logische Funktionen

Logische Funktionen sind nicht für die alleinige Verwendung in Alternativen gedacht. Für den Normal-Nutzer wird es aber der Hauptanwendungsbereich bleiben.

Funktion	Kurzbeschreibung
FALSCH	gibt den Wahrheitswert FALSCH zurück
NICHT	berechnet die logische Negation (Kehrwert)
ODER	verknüpft die Argumente (Wahrheitswerte) mittels logischer ODER-Funktion
UND	verknüpft die Argumente (Wahrheitswerte) mittels logischer UND-Funktion
WAHR	gibt den Wahrheitswert WAHR zurück
WENN	→ 3.8.2. echte Alternativen (WENN)

3.8.3. Mehrfachalternativen mit VERWEIS-Funktionen

Die Schachtelung von Alternativen stößt bei einer größeren Anzahl von Zweigen schnell an seine Grenzen. Genau so können häufige Änderungen der Bedingungen für die einzelnen Alternativen die Pflege der Kalkulation zum Alptraum werden lassen.

Hat man die Prüfbedingung in Tabellenform vorliegen, dann kann man zumeist die Funktionen SVERWEIS, WVERWEIS oder VERWEIS nutzen.

Zur Veranschaulichung wählen wir eine Rabatt-Berechnung für den Jahres-Umsatz von verschiedenen Kunden. Die Rabatt-Staffel liegt schon in Tabellen-Form vor. Weiterhin sind zu verschiedenen Kunden die Jahresumsätze aufgerechnet worden:

	A	B	C	D	E	F	G
1	Berechnung des Jahres-Rabatts						
2							
3		Rabatt-Staffel					
4		Umsatz	Jahres-Rabatt	Kunden-Status			
5		0	0%	'- - -			
6		1000	2%	Messing			
7		2000	3%	Bronze			
8		4000	4%	Silber			
9		8000	5%	Gold			
10		15000	6%	Platin			
11							
12	lfd. Nr.	Kunde	Umsatz	Rabatt [%]	Rabatt [€]	Status	
13	1	Meier GmbH	4729 €				
14	2	Just & Sohn	13094 €				
15	3	Heymann OHG	0 €				
16	4	Schnell & Gut	27450 €				
17	5	Schuhmann	186 €				
18	6	Fa. Kalk	9403 €				
19	7	Die Macher	394 €				
20							

Zum Ermitteln des Jahres-Rabatts müssen wir für jeden Kunden mit seinem Jahresumsatz in die Rabatt-Staffel-Tabelle gehen und den passenden %-Wert heraussuchen.

Für die Meier GmbH würden wir so auf 4% kommen. Mit diesem Wert könnten wir den Rabatt in Euro berechnen. Der (Kunden-)Status ergibt sich ebenfalls aus der Rabatt-Staffel und wäre bei der Meier GmbH "Silber".

In der nächsten Tabelle ist der Weg mit kurzen Arbeitshinweisen nochmals schrittweise aufgezeigt:

	A	B	C	D	E	F	G
1	Berechnung des Jahres-Rabatts						
2							
3		Rabatt-Staffel					
4		Umsatz	Jahres-Rabatt	Status			
5		0	0%	'- -'			
6		1000	2%	Messing			
7		2000	3%	Bronze			
8		4000	4%	Silber			
9		8000	5%	Gold			
10		15000	6%	Platin			
11							
12	lfd. Nr.	Kunde	Umsatz	Rabatt [%]	Raba		
13	1	Meier GmbH	4729 €				
14		n	13094 €				
15		DHG	0 €				
16		Gut	27450 €				
17		n	186 €				
18	6	Fa. Kalk	9403 €				
19	7	Die Macher	394 €				
20							

2. passenden Umsatz aus der 1. Spalte herausuchen

3. zugehörigen %-Wert aus der 2. Spalte holen

4. %-Wert in die Zelle eintragen

1. mit Umsatz des Kunden in die Rabatt-Staffel

In etwa so arbeitet auch die Funktion SVERWEIS. Eingegeben wird sie in die Zielzelle – also hier z.B. D13. Als Parameter übergeben wir als erstes den Wert, nach dem in der Such-Tabelle (hier Rabatt-Staffel) gesucht werden soll. In unserem Beispiel ist dies der Umsatz in Zelle C13. Als zweiten Parameter benötigt die SVERWEIS-Funktion die Ausdehnung der Such-Tabelle (Matrix). Dabei interessieren nur die reinen Daten. Für unser Beispiel ist es B5 bis D10 (C10 würde hier auch schon reichen). Der Suchwert muss dabei immer in der ersten Spalte dieser Tabelle stehen! Beachten Sie unbedingt, dass die Such-Tabelle im Konstanten-Bereich liegt od. anders gesagt, mit absoluten Adressen für die Such-Tabelle (Matrix: **\$B\$5: \$D\$10**) gearbeitet werden sollte.

Mit dem dritten Parameter wird mitgeteilt, aus welcher Spalte EXCEL den Ergebniswert holen soll. Die Zählung bezieht sich nur auf die Suchtabelle! In unserem Beispiel steht der %-Wert für den Jahres-Rabatt in der zweiten (2.) Spalte.

```
=SVERWEIS(C13; $B$5: $D$10; 2)
```

Zwischenwerte werden in der Such-Tabelle ebenfalls gefunden. Dazu ist es aber notwendig, dass die Tabelle aufsteigend sortiert ist. Will man in einer unsortierten Tabelle suchen oder dürfen nur die Tabellen-Werte benutzt werden, dann ist an die SVERWEIS-Funktion ein FALSCH als 4. (optionalen) Parameter zu übergeben. Der Wert WAHR für sortierte Folgen mit ungefährender Entsprechungssuche kann in jedem Fall weggelassen werden.

Der Syntax für die Funktion lautet:

```
=SVERWEIS( Suchwert ; Suchtabelle; Ergebnisspalte [; Suchoption ] )
```

Die Berechnung des Rabatts in Zelle E13 ist eine unkomplizierte Übung. Mit einfache Prozentrechnung ergibt sich:


=C13 * D13

Fehlt nur noch der Kundenstatus in Spalte F. Dafür nutzen wir die gleiche Funktion mit der gleichen Such-Tabelle. Lediglich der Spalten-Index ist jetzt **3**, da wir ja die dritte Spalte nutzen.

=SVERWEIS(C13; \$B\$5:\$D\$10; 3)

Letztendlich brauchen wir die Funktionen nur noch auf die unteren Zellen auszufüllen.

	A	B	C	D	E	F	G
1	Berechnung des Jahres-Rabatts						
2							
3		Rabatt-Staffel					
4		Umsatz	Jahres-Rabatt	Kunden-Status			
5		0	0%	'- - -			
6		1000	2%	Messing			
7		2000	3%	Bronze			
8		4000	4%	Silber			
9		8000	5%	Gold			
10		15000	6%	Platin			
11							
12	lfd. Nr.	Kunde	Umsatz	Rabatt [%]	Rabatt [€]	Status	
13	1	Meier GmbH	4729 €	4 %	189 €	Silber	
14	2	Just & Sohn	13094 €	5 %	655 €	Gold	
15	3	Heymann OHG	0 €	0 %	0 €	- - -	
16	4	Schnell & Gut	27450 €	6 %	1647 €	Platin	
17	5	Schuhmann	186 €	0 %	0 €	- - -	
18	6	Fa. Kalk	9403 €	5 %	470 €	Gold	
19	7	Die Macher	394 €	0 %	0 €	- - -	
20							



Soll die Nachschlage-Tabelle nicht in der Kalkulation zu sehen sein, dann kann man die Werte auch in die Funktion selbst aufnehmen. Dazu verwendet man die Verkettungs-Funktion. Innerhalb einer geschweiften Klammer werden die Zeilen (der Nachschlage-Tabelle) durch Semikolon getrennt. Die Elemente einer Zeile – also die Spaltenzellen – werden durch einen Punkt voneinander getrennt.

Für die obige Formel gilt dann:

=SVERWEIS(C13; {0.0%.“- - -“; 1000.2%.“Messing“; 2000.3%. “Bronze“; 4000.4%.“Silber“; 8000.5%.“Gold“; 15000.6%.“Platin“}; 3)

(**Blau+Rot**...umgesetzte Tabelle; **Rot**...eine ausgewählte Zeile)

Aufgabe A:

1. Erstellen Sie ein Rechnungs-Kalkulation für einen Super-Mini-Buchversand! Dieser bietet gerade einmal 7 Bücher an. In einer Tabelle sind die wichtigsten Buchdaten zusammengefasst (als separate Tabelle führen!):

Bestell-Nr.	Kurztitel	Lagerbestand	Verkaufs-Preis
152	Basiswissen EXCEL	23	39,00 €
281	Arbeitsheft für Anfänger	92	12,95 €
474	EXCEL-Tricks für Profi's	4	49,95 €
598	MÜLLER; ohne EXCEL?	17	19,00 €
629	Lösungen zum EXCEL-Arbeitsheft	3	27,95 €
745	SCHULZ, MEIER: EXCEL XP	13	79,95 €
913	EXCEL ohne Probleme	16	34,95 €

2. Die Rechnung (mindestens 3 Rechnungspositionen) soll auf alle Fälle die Anzahl, den Kurztitel, den Einzelpreis und die Preissummen enthalten. Bereiten Sie die Kalkulation und den Ausdruck so weit vor, dass Ihnen die Bestellung (Bestell-Nummer und Anzahl) nur noch "telefonisch" (vom Kursleiter) angesagt werden muss! Nur für diese Bestellung drucken Sie dann die Rechnung aus!

für die gehobene Anspruchsebene:

3. Weisen Sie die enthaltene Mehrwertsteuer für die Einzel-Posten und für die Rechnung insgesamt auf der Rechnung aus!

für FREAK's:

4. Verkaufen Sie maximal nur so viele Bücher, wie im Bestand sind!

Aufgabe B:

Der Sportverein "Sport frei" hat in der Satzung folgenden Modus zur Berechnung des Mitgliedsbeitrages festgelegt. Je nach Jahres-Einkommen ergibt sich ein konkreter Prozentsatz:

bis 10.000 Euro 1%, bis 20.000 Euro 1,5%, bis 30.000 Euro 2% und da rüber 3%.

Berechnen Sie die jeweiligen Beiträge (auf ganze Euro abgerundet) für folgende Mitglieder!

Name	Einkommen	Status
Meier	34.000 €	
Schreiber	300 €	Schüler
Schulz	14.000 €	
Wolff		Schüler

Name	Einkommen	Status
Jahn	4.000 €	Student
Schmidt	34.000 €	
Müller	13.000 €	
Voigt	6.000 €	

für die gehobene Anspruchsebene:

Schüler und Studenten zahlen pauschal 30 Euro pro Jahr.

Im Prinzip ist die Funktion WVERWEIS genau so benutzbar, wie die besprochene Funktion SVERWEIS. Einziger Unterschied ist die Anordnung der Such-Tabellendaten in Zeilen und nicht in Spalten. Logischerweise wird also als 3. Parameter die Indexzeile für das Ergebnis erwartet.

=WVERWEIS(Suchwert ; Suchtabelle; Ergebniszeile [; Suchoption])

	A	B	C	D	E	F
1	Cotton-Großhandel GmbH					
2						
3	Bestell-Nr.	A943	A288	C741	C391	
4	Bezeichnung	Hemd 1/1 Arm	Hemd 1/2 Arm	Hose	Hose, kurz	
5	Größe	M	XL	K	M	
6	Farbe	rot	weiß	blau	blau	
7	Preis	12,49 €	23,99 €	39,95 €	9,99 €	
8	Gewicht [kg]	0,2	0,15	0,55	0,4	
9						
10						
11	Bestell-Nr.	Anzahl	Einzel-Preis	Gesamt-Preis	Gesamt-Gewicht	
12	C391	250				
13	A288	375				
14			zusammen:			
15						

Schauen wir uns zuerst wieder die Arbeitsschritte an! Nachdem die Bestell-Nummer eingetragen wurde, muss für die Bereitstellung des Einzel-Preis (C12 bzw. C13) in der obigen Tabelle die richtige Spalte rausgesucht werden. In der Such-Tabelle wird in der Zeile 1 (ganz oben) nach der passenden Bestell-Nummer gesucht. Ist sie gefunden, wird in der 5. Zeile nachgeschlagen, welcher Preis vorgesehen ist.

	A	B	C	D	E	F
1	Cotton-Großhandel GmbH					
2						
3	Bestell-Nr.	A943	A288	C741	C391	
4	Bezeichnung	Hemd 1/1 Arm	Hemd 1/2 Arm	Hose	Hose, kurz	
5	Größe	M	XL	K	M	
6	Farbe	rot	weiß	blau	blau	
7	Preis	12,49 €	23,99 €	39,95 €	9,99 €	
8	Gewicht [kg]	0,2	0,15	0,55	0,4	
9						
10						
11	Bestell-Nr.	Anzahl	Einzel-Preis	Gesamt-Preis	Gesamt-Gewicht	
12	C391	250				
13	A288	375				
14			zusammen:			
15						

Leiten wir nun die entsprechenden WVERWEISE ab. Zuerst für den Einzel-Preis in Zelle C12. Der Suchwert kommt aus der Zelle **A12** und muss in der 1. Zeile der Such-Tabelle (Matrix) **B3:E8** nachgeschlagen werden (absoluten Zellbezug beachten!). Der Ergebniswert steht in der Zeile **5** der Matrix. Da keine Zwischenwerte zugelassen werden und die Bestellnummer auch nicht zwangsläufig sortiert sind, muss die zusätzliche Option **“Falsch“** mit angegeben werden. Somit ergibt sich die Formel:

```
=WVERWEIS(A12; $B$3:$E$8; 5; FALSCH)
```

Der Gesamt-Preis wird dann unter Verwendung von der Anzahl aus Zelle B12 und dem gerade ermittelten Einzel-Preis in Zelle C12 berechnet.

```
=B12 * C12
```

Um das Gesamt-Gewicht zu berechnen, nutzen wir die Funktion von C12 aus. Da nicht mehr der Preis sondern das Gewicht gefragt ist, greifen wir auf Zeile **6** zurück. Der zurückgelieferte Wert wird gleich mit der Anzahl aus Zelle **B12** multipliziert.

```
=WVERWEIS(A12; $B$3:$E$8; 6; FALSCH) * B12
```

Aufgaben:

1. **Vervollständigen Sie die Kalkulation für die Cotton-Großhandel GmbH!**
2. **Zur Erweiterung des Sortiments werden die folgenden Artikel in den Vertrieb aufgenommen. Erweitern Sie die Kalkulation so, dass auch maximal 5 Artikel mit beliebigen Anzahlen bestellt werden können!**

Bestell-Nr.	B166	B333	Z219	Z748	S832	K539	K538
Bezeichnung	Slip	BH	Strumpf	Socken	Strampler	Schuh	Schuh
Größe	4	90 C	36-40	32-36	104	38	39
Farbe	weiß	rot	braun	weiß	creme	braun	braun
Preis	3,95 €	9,99 €	2,75 €	2,50	4,90 €	23,95 €	23,95 €
Gewicht	0,05	0,1	0,075	0,05	0,15	0,35	0,36

Bleibt von den Verweis-Funktionen nur die Funktion VERWEIS über. Bei ihr können die Such- und Ergebniswerte unabhängig voneinander in Spalten oder Zeilen stehen. Der Verknüpfungspunkt ist die Positions-Nummer (Index) in der Such-Spalte/Zeile (allg. Vektor). Der Such- und der Ergebnis-Vektor müssen also gleich lang sein. Vielfach wird von erfahrenen EXCELern lieber die – im Syntax flexiblere – VERWEIS-Funktion benutzt. Der VERWEIS-Funktion werden statt der Suchtabelle nur die Suchspalte und die Ergebnisspalte jeweils als Zellbereich angegeben (Vektor-Version). Dagegen ist nichts zu sagen. Für eine automatische Spalten-/Zeilen-Auswahl (variable Ergebnisspalten bzw. –zeilen) sind aber nur die SVERWEIS- bzw. WVERWEIS-Funktionen geeignet. Neben der Vektor-Version gibt es eine Matrix-Version (Matrix-Syntax). Diese entscheidet selbstständig, ob die Suchkriterien in der ersten Spalte oder Zeile stehen ??? (abhängig von Form der Tabelle!). Hier ist immer ein gründlicher Test – auch nach Änderung der Tabellenstruktur – notwendig.

komplexe Aufgabe:

Realisieren Sie einen übersichtlichen (schrittweisen) ausdrückbaren Lohnberechnungsbogen! Damit die Berechnungen immer sehr schnell realisiert werden können, sollen die notwendigen Angaben (Eingaben) auf ein Minimum reduziert werden! Dem Hauptbuchhalter schwebt dabei vor nur noch den Nachnamen und die geleisteten Arbeitsstunden angeben zu müssen. Der Personalbestand soll maximal 15 Personen umfassen. Es gelten die nachfolgenden (idealisierten) Bedingungen!

1. Diskutieren Sie zuerst in einer kleinen Gruppe den grundsätzlichen Aufbau der Kalkulation! Überlegen Sie welche Funktionen und Techniken Sie anwenden wollen und beachten Sie deren Besonderheiten! Fertigen Sie eine Grobskizze an und notieren Sie benötigte Funktionen und Techniken – ev. mit Verweispfeilen – in der Skizze!

Personal-Nummer	Name	Vorname	Steuerklasse	Kinder	Stundenlohn (Brutto)
6927	Müller	Dirk	III	1	17,00 €
8835	Johanson	Sven	IV	1	19,50 €
9342	Merck	Simone	I	0	15,00 €
2746	Zander	Anette	IV	2	17,00 €
7344	Mendel	Johann	II	2	16,75 €

Lohnsteuer in Euro

bis Bruttogehalt	Steuerklasse				
	I	II	III	IV	V
500	40,00	-	30,00	-	-
750	45,00	40,00	35,00	-	-
1000	50,00	45,00	40,00	30,00	-
1250	62,50	57,50	47,50	37,50	-
1500	75,00	70,00	55,00	45,00	25,00
1750	87,50	82,50	62,50	52,50	32,50
2000	100,00	95,00	70,00	60,00	40,00
2250	130,00	125,00	80,00	70,00	60,00
2500	160,00	155,00	90,00	80,00	70,00
2750	190,00	185,00	100,00	90,00	80,00
3000	220,00	215,00	110,00	100,00	90,00

je Kind 10 % Steuerminderung auf die Gesamtlohnsteuer

weitere Abgaben in Prozent

	männlich	weiblich	je Kind
Krankenversicherung	13,00	10,50	1,00
Rentenversicherung	7,00	8,00	-
Arbeitslosenversicherung	3,00	4,50	-
Pflegeversicherung	3,50	5,00	-

2. Prüfen Sie an Ihrer fertigen Kalkulation, inwieweit Ihr Plan das Ergebnis vorausgenommen hat! Notieren Sie in einem (nicht auszudruckenden) Textfeld die aufgetretenen Problem und Veränderungen!
3. Für eine auszudruckende Gehaltsabrechnung wir Ihnen eine Person und deren geleistete Stundenzahl angesagt!